


# 6 Estruturas de Seleção


As estruturas de seleção, também conhecidas por estruturas de decisão, permitem a escolha de um grupo de ações para serem executadas de acordo com a avaliação de uma expressão ou um conjunto delas. Dessa forma, passamos a ter a execução condicional de um grupo de comando do programa.

## 6.1 Se

Em pseudolinguagem (Portugol), um bloco de comandos deve ser entendido como um conjunto de instruções delimitadas por abertura e fechamento de bloco. Na elaboração do fluxograma utilizamos o símbolo do losango para caracterizar o uso da estrutura de seleção:



Como exemplo, vamos considerar uma situação na qual devemos indicar a partir da idade de uma pessoa se a mesma é maior ou menor de idade. Dessa forma, no fluxograma poderíamos então realizar a seguinte representação:




Utilizando pseudolinguagem o comando de seleção **se** é representado da maneira indicada pelo trecho de programa mostrado a seguir. No trecho de programa a seguir devemos avaliar a expressão presente no comando **se**, ou seja, se o valor da variável idade for maior ou igual a 18 ( $idade \geq 18$ ). Caso a expressão seja avaliada como verdadeira a mensagem "Maior de idade" será exibida no console, caso contrário, será executada a instrução contida no **senao** e, desta maneira, a mensagem "Menor de idade" será mostrada no console.


```
se (idade >= 18)
    escreva("Maior de idade")
senao
    escreva("Menor de idade")
fimse
```

Utilizando Scratch temos o bloco "se então senão" para representar uma estrutura de seleção, conforme é mostrado na imagem abaixo:



Após esses conceitos iniciais aplicaremos, neste próximo exemplo, o conceito de estrutura de seleção dentro de um programa completo. Vamos considerar um programa no qual o usuário irá digitar o ano de nascimento e em seguida, a idade será calculada e, ao final, será informado se o mesmo é maior ou menor de idade. Podemos observar a representação da solução para o cálculo da idade por meio do fluxograma:





Utilizando pseudolinguagem teríamos, como uma possível solução para o problema, o programa a seguir:


```

algoritmo "Calcular idade"
var
  ano_nasc, idade:inteiro
inicio
  escreva("Digite o ano de nascimento: ")
  leia(ano_nasc)


  idade <- 2018 - ano_nasc


  se (idade >= 18) entao
    escreva("Maior de idade")
  senao
    escreva("Menor de idade")
  fimse
fimalgoritmo
  
```

Apresentamos também em Scratch a solução para este mesmo problema.



Em determinadas situações são necessários vários comandos **se** para resolvermos um determinado problema, por exemplo, vamos considerar a necessidade de a partir de três números digitados pelo usuário, identificarmos o maior deles. Neste caso, como são três as possibilidades, um único comando **se** não irá conseguir resolver o problema. Em situações como essa, devemos utilizar os comandos de modo aninhados, conforme ilustrado pelo fluxograma apresentado a seguir:





Adotando a pseudolinguagem a solução poderia ser expressa da seguinte maneira:

```
algoritmo "Maior número"  
var  
  a,b,c: inteiro  
inicio  
  escreva("Digite o primeiro valor: ")  
  leia(a)  
  escreva("Digite o segundo valor: ")  
  leia(b)  
  escreva("Digite o terceiro valor: ")  
  leia(c)  
  
  se ((a > b) e (a > c)) entao  
    escreva("O maior é ",a)  
  senao  
  se (b > c) entao  
    escreva("O maior é ",b)  
  senao  
    escreva("O maior é ",c)  
  fimse  
  fimse  
fimalgoritmo
```

Em Scratch podemos representar a solução para esse mesmo problema da seguinte maneira. Observe a forma como os blocos de montagem de "se" foram encadeados.

